

ATENÇÃO! As atividades práticas serão por meio de atividades já solucionadas. Nossa ideia é que você entenda o contexto e recrie o código-fonte apresentado em seu Portugol Studio. Com isso, você terá exemplos mais robustos para poder seguir sua evolução.

ATIVIDADE 1

Faremos a construção de um programa que fará a validação de um número. Sempre que o usuário digitar um número ímpar ele seguirá sendo executado. A partir do momento em que um número par for informado, o programa será encerrado. Importante informar que o código será criado de modo a ser encerrado apenas quando um número par for digitado, ou seja, ele pode rodar infinitamente ou até que o usuário se canse, enquanto a regra definida não for cumprida. Ao final de nosso programa, informaremos quantos números foram digitados até que ele fosse encerrado.

IMPORTANTE

Lembre-se que um número par é divisível por dois sem resto.

Por exemplo:

- *Números Pares – $28 / 2 = 14$ e seu resto é 0 (zero);*
- *Números Ímpar – $15 / 2 = 7$ e seu resto é 1 (um).*

Para se saber o resto de uma divisão usamos o símbolo % no lugar de /. Ou seja, ao invés de utilizarmos $28 / 2$ que é igual a 14, usamos $28 \% 2$ cujo resultado é 0.

Então:

- */ - representa que buscamos o RESULTADO de uma divisão;*
- *% - representa que queremos o RESTO de uma divisão.*

Figura 1 - Atividade 1

```
programa
{
    funcao inicio()
    {
        inteiro qtdNumeros=0, numDigitado
        logico numPar = verdadeiro

        faca{
            limpa()
            escreva("Digite um numero: ")
            leia(numDigitado)
            qtdNumeros++

        }enquanto(numDigitado%2==0)
        escreva("Você digitou um número ímpar, não tem direito de digitar mais números")
    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 2

Construiremos uma calculadora que nos permita fazer o cálculo das quatro operações básicas:

- Soma representado pelo sinal + (Ex.: $1 + 1 = 2$)
- Subtração representado pelo sinal – (Ex.: $3 - 2 = 1$)
- Multiplicação representado pelo sinal * (Ex.: $5 * 5 = 25$)
- Divisão representado pelo sinal / (Ex.: $15 / 3 = 5$)

Para isso, seguiremos os seguintes passos:

- Solicitação do primeiro valor;
- Solicitação da operação matemática (+, -, * ou /);
- Solicitação do segundo valor;
- Realização do cálculo;
- Perguntar se o usuário deseja prosseguir calculando:
 - o Se sim, o processo será iniciado já no passo 2, tendo o resultado no passo 4 como primeiro valor;
 - o Se não, será mostrado o resultado final:
 - Por exemplo, $4 + 3 = 7 * 4 = 28 / 14 = 2$.

Figura 2 - Atividade 2

```
programa
{
    inclui biblioteca Texto --> t

    funcao inicio()
    {
        real num1, num2, resultado
        cadeia continuar = ""
        logico operValido
        caracter oper

        limpa()
        resultado=0.0
        escreva("\nPrimeiro valor: ")
        leia(num1)
        faca {
            escreva("\nOperação (*,+,-,/): ")
            leia(oper)

            escreva("\nSegundo valor: ")
            leia(num2)

            escolha(oper) {
                caso '*':
                    resultado=num1*num2
                    pare

                caso '+':
                    resultado=num1+num2
                    pare

                caso '-':
                    resultado=num1-num2
                    pare

                caso '/':
                    se (num2==0){
                        escreva ("\nNão é possível efetuar divisão por zero!")
                    } senao {
                        resultado = num1/num2
                    }
                    pare

                caso contrario:

                    escreva("\nOperação inválida!")

            }

            num1 = resultado
            escreva("\nDeseja continuar? (S/N) ")
            leia(continuar)
            continuar = t.caixa_alta(continuar)
        } enquanto (continuar=="S")
        escreva("\n" + num1 + " " + oper + " " + num2 + " = " + resultado)

    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 3

E finalmente, faremos um algoritmo que solicitará ao usuário a nota que ele tirou em uma prova. A partir dela, faremos uma validação:

- Caso a nota seja maior do que 7, o aluno está aprovado;
- Caso a nota seja menor do que 7, o aluno será autorizado a realizar uma prova de recuperação.

A ideia é que o aluno possa fazer quantas provas de recuperação forem necessárias até ser aprovado. Ou seja, ele precisa atingir 7 ou mais em algum momento. Ao final do nosso programa, mostraremos se a pessoa precisou ou não de uma prova de recuperação e, se sim, quantas delas precisou fazer.

Figura 3 - Atividade 3

```
programa
{
    funcao inicio()
    {
        real nota
        inteiro qtdProvasRec
        qtdProvasRec = 0

        escreva("Digite sua nota: ")
        leia(nota)

        enquanto((nota < 7) ou (nota > 10)){
            limpa()

            se ((nota >= 0) e (nota < 7)){
                qtdProvasRec += 1
                escreva("Sua nota foi "+nota+", e você tem direito a recuperação!")
                escreva("\nDigite a nota tirada na recuperação: ")
                leia(nota)
            } senao {
                escreva("Sua nota foi inválida. Digite novamente: ")
                leia(nota)
            }
        }

        limpa()

        escreva("Parabéns! Você passou!")

        se (qtdProvasRec == 0) {
            escreva("\nVocê não precisou de prova de recuperação!")
        } senao se (qtdProvasRec == 1) {
            escreva("\nVocê precisou apenas de uma prova de recuperação!")
        } senao {
            escreva("\nVocê precisou de " + qtdProvasRec + " provas de recuperação!")
        }
    }
}
```

Fonte: Autores, 2023.