

ATENÇÃO! As atividades práticas serão por meio de atividades já solucionadas. Nossa ideia é que você entenda o contexto e recrie o código-fonte apresentado em seu Portugol Studio. Com isso, você terá exemplos mais robustos para poder seguir sua evolução.

ATIVIDADE 1

Construiremos um algoritmo que mostrará na tela 10 vezes a seguinte frase espirituosa:

“A careca do Silvio brilha muito. Muito brilha a careca do Silvio.”

Figura 1 - Atividade 1

```
programa
{
    funcao inicio()
    {
        para (inteiro i=0; i<10; i++)
        {
            escreva("Frase " + (i+1) + ": A careca do Silvio brilha muito. Muito brilha a careca do Silvio.\n")
        }
    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 2

Criaremos um algoritmo capaz de solicitar ao usuário 10 (dez) números inteiros. Em seguida, será feita a separação desses números entre pares e ímpares. Ao final, mostraremos a quantidade e a soma dos números pares e também dos ímpares.

Figura 2 - Atividade 2

```
programa
{
    funcao inicio()
    {
        inteiro numero

        inteiro qtdPar = 0
        inteiro somaPar = 0

        inteiro qtdImpar = 0
        inteiro somaImpar = 0

        escreva("Este programa foi criado para fazer a contagem e a soma de números, separando-os entre pares e ímpares.\n")

        para (inteiro i=0; i<10; i++)
        {
            escreva("Insira o " + (i+1) + "º número: ")
            leia(numero)

            se((numero % 2) == 0)
            {
                qtdPar = qtdPar + 1
                somaPar = somaPar + numero
            } senao {
                qtdImpar = qtdImpar + 1
                somaImpar = somaImpar + numero
            }
        }

        limpa()

        escreva("Dos 10 números informados, temos o seguinte:\n\n")
        escreva("Quantidade de números pares: " + qtdPar + "\n")
        escreva("Soma dos números pares: " + somaPar + "\n\n")
        escreva("Quantidade de números ímpares: " + qtdImpar + "\n")
        escreva("Soma dos números ímpares: " + somaImpar + "\n")
    }
}
```

Fonte: Autores, 2023.