

ATENÇÃO! As atividades práticas serão por meio de atividades já solucionadas. Nossa ideia é que você entenda o contexto e recrie o código-fonte apresentado em seu Portugol Studio. Com isso, você terá exemplos mais robustos para poder seguir sua evolução.

ATIVIDADE 1

Faremos um programa que trará o nome de um mês a partir de seu respectivo número. Por exemplo, mês 1 é janeiro, mês 2 é fevereiro, e assim por diante. Assim, perguntaremos ao usuário um o número de 1 a 12 e, assim, ele saberá a qual mês ele corresponde. Caso o usuário digite um número fora do intervalo entre 1 e 12, ele receberá uma mensagem de que está fora do intervalo.

Figura 1 - Atividade 1

```
programa
{
    funcao inicio()
    {
        inteiro nome;

        escreva("Digite um número correspondente a um mês: ")
        leia(nome)

        limpa()

        escolha(nome){

            caso 1:
                escreva("Janeiro")
                pare

            caso 2:
                escreva("Fevereiro")
                pare

            caso 3:
                escreva("Março")
                pare

            caso 4:
                escreva("Abril")
                pare

            caso 5:
                escreva("Maio")
                pare

            caso 6:
                escreva("Junho")
                pare

            caso 7:
                escreva("Julho")
                pare

            caso 8:
                escreva("Agosto")
                pare

            caso 9:
                escreva("Setembro")
                pare

            caso 10:
                escreva("Outubro")
                pare

            caso 11:
                escreva("Novembro")
                pare

            caso 12:
                escreva("Dezembro")
                pare

            caso contrario:
                escreva("Mês inválido")
        }
    }
}
```

Fonte: Autores, 2023.

ATIVIDADE 2

A prefeitura de Chuville quer implantar um sistema de rodízio de veículos, a fim de diminuir os congestionamentos da cidade. Para isso solicitou que fosse criado um programa que verificasse e definisse o dia que uma pessoa pode circular com seu veículo, a partir de sua placa. Para isso, construiremos um algoritmo que pede ao usuário o último número da placa de seu veículo e, a partir disso, informará o dia da semana que ele não pode circular nas ruas. Como regra, será tomada a seguinte definição em tabela:

Tabela 1 - Regras de negócio da Atividade 2

Dia da semana	Placas que não podem circular
Segunda-feira	0 e 5
Terça-feira	1 e 6
Quarta-feira	2 e 7
Quinta-feira	3 e 8
Sexta-feira	4 e 9

Fonte: Autores, 2023.

Figura 2 - Atividade 2

```

programa
{
    funcao inicio()
    {
        inteiro ultimoNumeroPlaca

        escreva("Qual o último número da placa?: ")
        leia(ultimoNumeroPlaca)

        escolha(ultimoNumeroPlaca){
            caso 0:
                escreva("Segunda-Feira")
                pare

            caso 1:
                escreva("Terça-Feira")
                pare
            caso 2:
                escreva("Quarta-Feira")
                pare

            caso 3:
                escreva("Quinta-Feira")
                pare
            caso 4:
                escreva("Sexta-Feira")
                pare

            caso 5:
                escreva("Segunda-Feira")
                pare

            caso 6:
                escreva("Terça-Feira")
                pare

            caso 7:
                escreva("Quarta-Feira")
                pare

            caso 8:
                escreva("Quinta-Feira")
                pare

            caso 9:
                escreva("Sexta-Feira")
                pare

            caso contrario:
                escreva("Valor inválido!")

        }
    }
}
    
```

Fonte: Autores, 2023.

ATIVIDADE 3

As Lojas VestADS querem controlar suas vendas. Para isso, construiremos um programa que permitirá ao vendedor inserir o valor de compra do produto, bem como a forma de pagamento a ser utilizada. Para isso, seguiremos algumas regras para calcular o valor final dessa compra. São elas:

Tabela 2 - Regras de negócio da Atividade 3

CÓDIGO	FORMA DE PAGAMENTO	REGRA DE NEGÓCIO
1	Cheque	- Usuário deverá selecionar se é a vista ou à prazo - Não há alteração de valor
2	Cartão de crédito	- Usuário deverá selecionar se é a vista ou à prazo - Se for a vista, não há alteração de valor final - Se for a prazo, acrescentará 5% no valor final - Parcelamento máximo é de 10x
3	Cartão de débito	- Há um desconto de 5% do valor final
4	Dinheiro / PIX	- Há um desconto de 10% do valor final

Fonte: Autores, 2023.

Após o cálculo, apresentaremos ao vendedor da loja: valor do produto, forma de pagamento, tipo do pagamento (a prazo ou a vista) e o valor final de venda.

Figura 3 - Atividade 3 (parte 1)

```
programa
{

    inclui biblioteca Matematica

    funcao inicio()
    {
        real valorCompras, valorFinal
        inteiro opcaoPag, opcaoParcPag, qtdParc
        cadeia formaPag, parcPag

        valorCompras = 0.0
        valorFinal = 0.0
        opcaoPag = 0
        opcaoParcPag = 0
        qtdParc = 0
        formaPag = ""
        parcPag = ""

        escreva("Bem-vindo ao sistema de vendas das Lojas VestADS!\n")
        escreva("Insira o valor das compras: R$ ")
        leia(valorCompras)

        faca
        {
            escreva("Selecione:\n")
            escreva("1 - Cheque\n")
            escreva("2 - Cartão de crédito\n")
            escreva("3 - Cartão de débito\n")
            escreva("4 - Dinheiro / PIX\n")
            escreva("\n")
            leia(opcaoPag)
            limpa()

            se ((opcaoPag < 1) ou (opcaoPag > 4))
            {
                escreva("OPÇÃO INVÁLIDA\n")
            }
        } enquanto((opcaoPag < 1) ou (opcaoPag > 4))
    }
}
```

Fonte: Autores, 2023.

Figura 4 - Atividade 3 (parte 2)

```
escolha(opcaoPag)
{
    caso 1:
        formaPag = "Cheque"
        valorFinal = valorCompras
        parcPag = "À vista"
        qtdParc = 1

        faca
        {
            escreva("A compra é:\n")
            escreva("1 - À vista\n")
            escreva("2 - À prazo (2x)\n")
            leia(opcaoParcPag)
            limpa()

            se ((opcaoParcPag < 1) ou (opcaoParcPag > 2))
            {
                escreva("OPÇÃO INVÁLIDA\n")
            }
        } enquanto((opcaoParcPag < 1) ou (opcaoParcPag > 2))

        escolha(opcaoParcPag)
        {
            caso 1:
                parcPag = "À vista"
                qtdParc = 1
                pare
            caso 2:
                parcPag = "À prazo"
                qtdParc = 2
                pare
        }
    pare
}
```

Fonte: Autores, 2023.

Figura 5 - Atividade 3 (parte 3)

```
caso 2:
    formaPag = "Cartão de crédito"

    faca
    {
        escreva("A compra é:\n")
        escreva("1 - À vista\n")
        escreva("2 - À prazo\n")
        leia(opcaoParcPag)
        limpe()

        se ((opcaoParcPag < 1) ou (opcaoParcPag > 2))
        {
            escreva("OPÇÃO INVÁLIDA\n")
        }
    } enquanto((opcaoParcPag < 1) ou (opcaoParcPag > 2))

    se (opcaoParcPag == 1)
    {
        valorFinal = valorCompras
        qtdParc = 1
        parcPag = "À vista"
    } senao {
        valorFinal = valorCompras + (valorCompras * 0.05)
        parcPag = "À prazo"

        faca
        {
            escreva("Informe a quantidade de parcelas (máximo de 10x):\n")
            leia(qtdParc)
            limpe()

            se ((qtdParc < 2) ou (qtdParc > 10))
            {
                escreva("OPÇÃO INVÁLIDA\n")
            }
        } enquanto((qtdParc < 2) ou (qtdParc > 10))
    }
pare
```

Fonte: Autores, 2023.

Figura 6 - Atividade 3 (parte 4)

```
    caso 3:
        formaPag = "Cartão de débito"
        valorFinal = valorCompras - (valorCompras * 0.05)
        parcPag = "À vista"
        qtdParc = 1
        pare

    caso 4:
        formaPag = "Dinheiro / PIX"
        valorFinal = valorCompras - (valorCompras * 0.10)
        parcPag = "À vista"
        qtdParc = 1
        pare
}

limpa()
escreva("DADOS DA COMPRA")
escreva("\nValor comprado: R$" + Matematica.arredondar(valorCompras, 2))
escreva("\nForma de pagamento: " + formaPag)
escreva("\nQuantidade de parcelas: " + parcPag + " - Em " + qtdParc + "x")
escreva("\nValor final da compra: R$ " + Matematica.arredondar(valorFinal, 2))
}
```

Fonte: Autores, 2023.